NAG Library Routine Document

G05MBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05MBF generates a vector of pseudorandom integers from the discrete geometric distribution with probability p of success at a trial.

2 Specification

```
SUBROUTINE G05MBF(MODE, P, N, X, IGEN, ISEED, R, NR, IFAIL)
INTEGER           MODE, N, X(N), IGEN, ISEED(4), NR, IFAIL
double precision  P, R(NR)
```

3 Description

G05MBF generates n integers x_I from a discrete geometric distribution, where the probability of $x_i = I$ (a first success after I trials) is

$$P(x_i = I) = p \times \left[(1-p)^{(I-1)}\right], \qquad I = 1, 2, \ldots.$$

The variates can be generated with or without using a search table and index. If a search table is used then it is stored with the index in a reference vector and subsequent calls to G05MBF with the same parameter value can then use this reference vector to generate further variates. If the search table is not used (as recommended for small values of p) then a direct transformation of uniform variates is used.

One of the initialization routines G05KBF (for a repeatable sequence if computed sequentially) or G05KCF (for a non-repeatable sequence) must be called prior to the first call to G05MBF.

4 References

Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison–Wesley

5 Parameters

1: MODE – INTEGER *Input*

On entry: a code for selecting the operation to be performed by the routine.

MODE $= 0$
: Set up reference vector only.

MODE $= 1$
: Generate variates using reference vector set up in a prior call to G05MBF.

MODE $= 2$
: Set up reference vector and generate variates.

MODE $= 3$
: Generate variates without using the reference vector.

Constraint: MODE $= 0$, 1, 2 or 3.

2: P – ***double precision*** *Input*

On entry: the parameter p of the geometric distribution representing the probability of success at a single trial.

Constraint: ***machine precision*** $\leq \mathrm{P} \leq 1.0$ (see X02AJF).

3: N – INTEGER *Input*

On entry: n, the number of pseudorandom numbers to be generated.

Constraint: $\mathrm{N} \geq 1$.

4: X(N) – INTEGER array *Output*

On exit: the n pseudorandom numbers from the specified geometric distribution.

5: IGEN – INTEGER *Input*

On entry: must contain the identification number for the generator to be used to return a pseudorandom number and should remain unchanged following initialization by a prior call to G05KBF or G05KCF.

6: ISEED(4) – INTEGER array *Communication Array*

On entry: contains values which define the current state of the selected generator.

On exit: contains updated values defining the new state of the selected generator.

7: R(NR) – ***double precision*** array *Communication Array*

On entry: if $\mathrm{MODE} = 1$, the reference vector from the previous call to G05MBF.

On exit: the reference vector.

8: NR – INTEGER *Input*

On entry: the dimension of the array R as declared in the (sub)program from which G05MBF is called.

Suggested value: $\mathrm{NR} = 6 + 42/\mathrm{P}$ approximately (for optimum efficiency in G05MZF).

Constraints:

if $\mathrm{MODE} = 0$ or 2, $\mathrm{NR} \geq 30/\mathrm{P} + 6$;
if $\mathrm{MODE} = 1$, NR should remain unchanged from the previous call to G05MBF;
if $\mathrm{MODE} = 3$, R is not referenced..

9: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: $\mathrm{IFAIL} = 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, N < 1.

IFAIL $= 2$

On entry, NR $< 30/\mathrm{P} + 6$ and MODE $= 0$ or 2.

IFAIL $= 3$

On entry, P $<$ ***machine precision***,
or P > 1.0.

IFAIL $= 4$

On entry, MODE $\neq 0$, 1, 2 or 3.

IFAIL $= 5$

MODE $= 0$ or 2 and P is so small that NR would have to be larger than the largest representable integer. Use MODE $= 3$ in this case.

IFAIL $= 6$

P is not the same as when R was set up in a previous call with MODE $= 0$ or 2.

7 Accuracy

Not applicable.

8 Further Comments

The time taken to set up the reference vector, if used, increases with the length of array R. However, if the reference vector is used, the time taken to generate numbers decreases as the space allotted to the index part of R increases. There is a point, depending on the distribution, where this improvement becomes very small and the recommended values for the length of array R in other routines are designed to approximate this point.

If P is very small then the storage requirements for the reference vector and the time taken to set up the reference vector becomes prohibitive. In this case it is recommended that the reference vector is not used. This is achieved by selecting MODE $= 3$.

9 Example

This example prints five pseudorandom integers from a geometric distribution with parameter $p = 0.001$, generated by a single call to G05MBF, after initialization by G05KBF.

9.1 Program Text

```
*     G05MBF Example Program Text
*     Mark 20 Release. NAG Copyright 2001.
*     .. Parameters ..
      INTEGER          NOUT, N, NR
      PARAMETER        (NOUT=6,N=10,NR=40000)
*     .. Local Scalars ..
      DOUBLE PRECISION P
      INTEGER          IFAIL, IGEN
```

```
*     .. Local Arrays ..
      DOUBLE PRECISION R(NR)
      INTEGER          ISEED(4), X(N)
*     .. External Subroutines ..
      EXTERNAL         G05KBF, G05MBF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G05MBF Example Program Results'
      WRITE (NOUT,*)
*     Set the distribution parameter P
      P = 0.001D0
*     Initialize the seed to a repeatable sequence
      ISEED(1) = 1762543
      ISEED(2) = 9324783
      ISEED(3) = 423441
      ISEED(4) = 742355
*     IGEN identifies the stream.
      IGEN = 1
      CALL G05KBF(IGEN,ISEED)
*     Choose MODE = 3 because P is close to 0
      IFAIL = 1
      CALL G05MBF(3,P,N,X,IGEN,ISEED,R,NR,IFAIL)
*
      IF (IFAIL.EQ.0) THEN
*
         WRITE (NOUT,99999) X
*
      ELSE
         WRITE (NOUT,99998) IFAIL
      END IF
*
99999 FORMAT (1X,I12)
99998 FORMAT (1X,' ** G05MBF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 G05MBF Example Program Results

         3699
          631
           29
         1228
          481
          341
         1394
          915
          332
          487
```